AD-A065 418 NAVY UNDERWATER SOUND LAB NEW LONDON CONN F/G 20/1

CONTINUOUS GRADIENT RAY TRACING SYSTEM (CONGRATE) BOTTOM LOSS M--ETC(U)

FEB 70 J S COHEN, T A GARRETT

UNCLASSIFIED USL-TM-2070-35-70 NL

1 OF 1

AD A065418



END

DATE FILMED

4-79

DDC

Code

Copy

USL Problem No.
A-001-11-00

Navy Underwater Sound Laboratory
New London, Connecticut 06320

CONTINUOUS GRADIENT RAY TRACING SYSTEM (CONGRATS) BOTTOM LOSS MODELS

by

Jeffrey S. Cohen and Thelda A. Garrett

USL Technical Memorandum No. 2070-35-70

5 February 1970

USL-TM-2070-35-70

40p.

INTRODUCTION

CONGRATS (CONtinuous Gradient Ray Tracing System) is an integrated collection of ray tracing programs designed to construct ray diagrams, to determine multipath eigenrays, to model acoustic propagation, to compute boundary and volume reverberation, and to solve numerous additional problems in underwater acoustics. The CONGRATS series has been documented in references (a), (b), and (c).

Bottom reflection loss is often an important factor in acoustic field predictions. The foundational CONGRATS programs, S0990 and S0991, require an input table of bottom loss in decibels as a function of grazing angle in degrees. This data is often difficult to obtain directly. Hence, various expressions have been developed to model bottom loss versus grazing angle.

If a predetermined table is not supplied, bottom loss values may be computed automatically by one of four models which have been added to the CONGRATS programs, S0990 and S0991. The first model, which depends on source frequency, is a least-squares fit of a fourth degree polynomial surface to modified AMOS bottom loss data. The model was developed by Dr. E. S. Eby (see reference (d)). The second model, supplied to the authors by E. M. Podeszwa, is an approximation to average bottom losses in MGS Acoustic Provinces 1, 2, 3, 4, and 5. The third and fourth models are empirical bottom reflection loss expressions, depending on both source frequency and bottom porosity, developed by NEL (described in reference (e)) and NUWC (see reference (f)), respectively. This memorandum will describe the four models, make simple comparisons, describe the input options needed to use the various models, and provide a listing of the revised Subroutines INPUT and BLOSS and the new Subroutine BTMLOS.

DISCUSSION OF MODELS

BOTTOM LOSS TABLES

When an acoustic ray reflects from the ocean bottom, the intensity of the ray diminishes. The amount of power lost depends on the frequency of the ray, the angle with which the ray intersects the bottom, and the characteristics of the bottom. If the bottom characteristics are assumed to be constant over the area of interest and if the frequency of each ray is assumed to be the source frequency, the bottom loss depends on grazing angle alone. In order to calculate bottom loss for each ray, the CONGRATS programs require a table of bottom loss values for grazing angles varying from 0° to 90° in increments of 1°. Linear interpolations are performed between these values in order to compute bottom loss for other grazing angles. This table may be supplied by the user on eleven computer cards. The first card contains the words "BOTTOM" and "LOSS" starting in columns 1 and 11, respectively (see Fig. 1(a)). Ten cards containing the 91 necessary values of bottom loss in decibels must follow the BOTTOM LOSS card (see reference (a)). The use of this table allows the bottom loss curve to be completely arbitrary.

AMOS BOTTOM LOSS

If the user is not familiar with the bottom characteristics but does know the frequency at which he is operating, the bottom loss table can be generated by the least-squares fit to modified AMOS data described in reference (d). This fit is good only for grazing angles less than 60°, however. Hence, an approximation to bottom loss data (presented in reference (d)) was developed by the authors for grazing angles steeper than 60°. The combination of this high-angle approximation and the least-squares fit was programmed for the CONGRATS series. The equations used are shown in Table 1. In order to implement the use of this model, the card shown in Fig. 1(b) must be used instead of the BOTTOM LOSS set described in reference (a). It should be noted that a FREQUENCY card (see reference (a)), such as the one in Fig. 2, must precede this BOTTOM LOSS card. Figure 3 is a graph of the approximated AMOS bottom loss versus grazing angle for frequencies of 250, 500, 1000, 2000, and 4000 Hz. The general trend shown in the plot is that bottom loss increases with frequency.

MGS BOTTOM LOSS

If the bottom losses in the operational area of interest are known to be similar to those of one of the MGS Acoustic Provinces, an approximation to the average bottom losses in Province 1, 2, 3, 4, or 5 can be supplied by the CONGRATS programs. The mathematical expressions used as the approximations are shown in Table 2. These formulas depend only on grazing angle, so bottom loss tables can be supplied by the programs with no additional data. The approximate bottom losses of MGS Acoustic Province 1, 2, 3, 4, or 5 will be computed if the BOTTOM LOSS card shown in Fig. 1(c), (d), (e), (f), or (g), respectively, appears in the data deck. The bottom losses of the five MGS Acoustic Provinces are plotted in Fig. 4. For typical sonars in use today, Provinces 1 and 2 provide excellent to good bottom reflectivity, Province 3 provides a fair bottom bounce area, and Provinces 4 and 5 are marginal to unusable.

NEL BOTTOM LOSS

The composition of the bottom, which can be usefully classified by the single parameter porosity, is an important factor in determining bottom reflectivity. Table 3 (reproduced from reference (f)) presents porosity as a function of bottom composition, with the bottom varying from fine clay (high bottom loss) to very coarse sand (high bottom reflectivity). The bottom loss model developed by the Naval Electronics Laboratory (and described in reference (e)) depends on both source frequency and bottom porosity, as well as grazing angle. The equation used is presented in Table 4. Figure 5 is a plot of NEL bottom loss versus grazing angle for frequencies of 250, 500, 1000, 2000, and 4000 Hz and a constant porosity of 0.60. The bottom loss increases with frequency in a fashion similar to that shown in the AMOS approximation. Figure 6 is a plot of NEL bottom loss versus grazing angle for porosities varying from 0.30 to 0.90 in steps of 0.15 and a constant frequency of 1 kHz. As bottom porosity increases, bottom loss also increases. Figure 1(h) presents the data card required to cause the NEL bottom loss equation to be used by the CONGRATS series for a porosity of 0.60 and a previously determined frequency.

NUWC BOTTOM LOSS

The Naval Undersea Warfare Center has developed a newer bottom loss expression, based primarily on the AMOS results, which also depends on both source frequency and bottom porosity (see reference (f)). The equation is presented in Table 4, below the equation used by NEL. The frequency factor, ABTMLOS, is a function consisting of straight line segments joining eight specified points. The eight points are tabulated in Fig. 7 and a plot of the function is drawn. The NUWC bottom loss increases with frequency and porosity (shown in Figs. 8 and 9, respectively) as did the NEL bottom loss. If the card shown in Fig 1(i) appears in the data deck, the NUWC bottom loss model is used by the CONGRATS programs for a porosity of 0.40 and a previously inputted frequency.

COMPARISON

For a frequency of 3.5 kHz the bottom loss curves derived from the AMOS, NEL, and NUWC models (with a porosity of 0.40 used in the NEL and NUWC approximations) are plotted in Fig. 10. If source frequency is the only information available, the AMOS model produces a reasonable approximation to a characteristic bottom loss curve. If a small (large) bottom porosity (and hence high (low) bottom reflectivity) is used in the NEL or NUWC equations, the bottom loss curve produced, at any given frequency, is lower (higher) than the curve produced by the AMOS approximation, which was developed for an average bottom. This phenomenon is demonstrated for a low porosity in Fig. 10. For a given frequency and porosity, the NEL model usually gives higher bottom loss curves than the NUWC model, especially for smaller grazing angles. The general shape and relationship of the two curves is shown in Fig. 10. If no other criterion is available for choosing between the NEL and NUWC bottom loss expressions, it should be noted that the NUWC model was developed more recently and is more widely accepted. If the operating area has been previously classified as an NGS Acoustic Province, the CONGRATS approximations to the average bottom losses in Provinces 1 - 5 can be considered quite reliable.

DISCUSSION OF INPUT

The choice of bottom loss model depends on the units field (the ten-column alphanumeric field starting in column 21 -- see reference (a)) of the BOTTOM LOSS card appearing in the CONGRATS data deck. A list of the nine words that may be used in the units field, a brief description of the model implemented by each, and a reference to a sample card are shown in Table 5. If the AMOS, NEL, or NUWC model is used, a FREQUENCY card (see reference (a)), such as the one shown in Fig. 2, must precede the BOTTOM LOSS card. The ten-column numeric field starting in column 31 of the BOTTOM LOSS card contains the porosity of the bottom if the NEL or NUWC expressions are used.

An option has been provided to input a minimum and/or maximum bottom loss value in decibels to be accepted by the CONGRATS programs. A bottom loss value calculated to be smaller (larger) than the minimum (maximum) bottom loss is converted to the minimum (maximum) value. The ten-column numeric fields starting in columns 41 and 51 of the BOTTOM LOSS card contain the minimum and maximum, respectively. If these fields are left blank a minimum of 0 dB and a maximum of 500 dB will be assumed. This option cannot be used if a bottom loss table is inputted.

SUMMARY

The fundamental CONGRATS programs, S0990 and S0991, have been revised to calculate bottom loss as a function of grazing angle. The bottom loss curves have been computed with frequency, MGS Acoustic Province, and both frequency and porosity as basic parameters. The bottom loss models have been described in detail, examples and comparisons have been provided, and the new input options have been explained. Subroutines INPUT and BLOSS have been revised in order to accommodate the new bottom loss models, and a new subroutine, Subroutine BTMLOS, has been written to compute the bottom loss curves. These three routines are listed in the appendix. The ray plotting, eigenray generation, and eigenray processing functions of the CONGRATS series have not been altered in any way other than as described in this memorandum. It is hoped that the additional bottom loss options will be a convenience for users of the CONGRATS series.

Jeffrey S. Cohen
JEFFREY S. COHEN
Mathematician

Thelda A. Garrett
THELDA A. GARRETT
Math Aid

REFERENCES

(a) H. Weinberg, "CONGRATS I: Ray Plotting and Eigenray Generation," USL Report No. 1052, 30 October 1969.

(b) J. S. Cohen and L. T. Einstein, "CONGRATS II: Eigenray Processing Programs," USL Report No. 1069, 5 February 1970.

(c) J. S. Cohen and T. A. Garrett, "CONGRATS Temperature and Salinity Sound Speed Conversion," USL Technical Memorandum No. 2070-412-69, 10 November 1969.

(d) E. S. Eby and B. W. Perneski, "Least-Squares Fit To Bottom Loss As A Function Of Frequency And Grazing Angle," USL Technical Memorandum Ser. No. 921-055-62, 25 September 1962.

(e) Sonar Simulation Computer Programs Phase I Interim Technical Report, Vol. 1, Contract N00140-68-C-0372, 21 November 1968, General Electric Company, Heavy Military Electronics Systems, Syracuse, New York.

(f) H. R. Hall and W. H. Watson, "An Empirical Bottom Reflection Loss Expression For Use In Sonar Range Prediction," July 1967, Naval Undersea Warfare Center, San Diego, California.

TABLE 1
AMOS Equations

Let f = frequency(kHz), Θ = Grazing Angle(deg), N_B = Bottom Loss(dB)

<u>If $f \leq 1$ kHz,</u>

for $\Theta = 0^o$: $N_B = 0.0$

for $1^o \leq \Theta \leq 14^o$:
$$N_B = 2.96026 + 1.26990F + 1.60430\phi + .25257F^2 + .66161F\phi + .06072\phi^2 + .02000F^3 + .09833F^2\phi - .01888F\phi^2 - .02619\phi^3 - .00303F^4 + .00778F^3\phi - .00833F^2\phi^2 - .01250F\phi^3 - .01310\phi^4$$
where $\phi = (\Theta - 10)/5$ and $F = \log_2 f$

for $15^o \leq \Theta \leq 39^o$:
$$N_B = 4.35024 + 1.17091F + .46874\phi + .08272F^2 + .17115F\phi - .07547\phi^2 - .00386F^3 + .01295F^2\phi - .01678F\phi^2 - .00555\phi^3 - .00003F^4 - .00193F^3\phi - .00074F^2\phi^2 - .00224F\phi^3 + 0\cdot\phi^4$$
where $\phi = (\Theta - 30)/5$ and $F = 2 \log_2 f + 3$

for $40^o \leq \Theta \leq 60^o$:
$$N_B = 4.84820 + 1.46858F - .11457\phi + .11036F^2 - .01238F\phi - .00953\phi^2 - .01000F^3 - .00375F^2\phi - .00929F\phi^2 + .00833\phi^3 + 0\cdot F^4 - .00042F^3\phi - .00268F^2\phi^2 + .00830F\phi^3 - .00208\phi^4$$
where $\phi = (\Theta - 50)/5$ and $F = 2 \log_2 f + 3$

for $61^o \leq \Theta \leq 90^o$:
$$N_B = 4.61430 + 1.47306F - .047516\phi + .09214F^2 - .01084F^3 - .012717F\phi$$
where $\phi = \Theta - 60$ and $F = 2 \log_2 f + 3$

NOTE: For $61^o \leq \Theta \leq 90^o$ N_B can also be written:
$N_B = N_B(\Theta - 1) - .085667 - .025434F$
where $F = \log_2 f$

TABLE 1 (Continued)
AMOS Equations

<u>If $f > 1$ kHz,</u>

for $\theta = 0^{o}$:

$$N_B = 0.0$$

for $1^{o} \leq \theta \leq 14^{o}$:

$$N_B = 2.96026 + 1.26990F + 1.60430\phi + .25257F^2 + .66161F\phi + .06072\phi^2 + .02000F^3 + .09833F^2\phi - .01888F\phi^2 - .02619\phi^3 - .00303F^4 + .00778F^3\phi - .00833F^2\phi^2 - .01250F\phi^3 - .02310\phi^4$$

where $\phi = (\theta - 10)/5$ and $F = \log_2 f$

for $15^{o} \leq \theta \leq 39^{o}$:

$$N_B = 12.25083 + 3.88808F + 1.14802\phi + .16312F^2 + .04386F\phi - .18835\phi^2 - .02857F^3 - .08214F^2\phi - .05952F\phi^2 + .00667\phi^3 + .01071F^4 - .00179F^3\phi + .00476F^2\phi^2 + .01389F\phi^3 - .00348\phi^4$$

where $\phi = (\theta - 30)/5$ and $F = \log_2 f - 1$

for $40^{o} \leq \theta \leq 60^{o}$:

$$N_B = 13.59067 + 3.74190F - .25928\phi + .13266F^2 - .03764F\phi - .08816\phi^2 - .01333F^3 + .01214F^2\phi - .00959F\phi^2 + .00500\phi^3 - .02000F^4 - .00167F^3\phi + .00867F^2\phi^2 - .00167F\phi^3 + .00083\phi^4$$

where $\phi = (\theta - 50)/5$ and $F = \log_2 f - 1$

for $61^{o} \leq \theta \leq 90^{o}$:

$$N_B = 12.77275 + 3.61490F - .126335\phi + .19162F^2 - .000333F\phi - .01667F^3 + .034002\phi/(F + 1) - .02000F^4$$

where $\phi = \theta - 60$ and $F = \log_2 f - 1$

NOTE: For $61^{o} \leq \theta \leq 90^{o}$ N_B can also be written:
$N_B = N_B(\theta - 1) - .126002 - .000333F + .034002/F$
where $F = \log_2 f$

TABLE 2
MGS Equations

Let θ = Grazing Angle(deg) and N_B = Bottom Loss(dB)

for Province 1:
$$N_B = 2.2435 \log_e(.1260\theta + 1.496)$$

for Province 2:
$$N_B = 3.4315 \log_e(.1056\theta + 2.842)$$

for Province 3:
$$N_B = 2.4910 \log_e(.8864\theta + 10.526)$$

for Province 4:
$$N_B = 2.8377 \log_e(1.8754\theta + 15.685)$$

for Province 5:
$$N_B = 2.4036 \log_e(20.5760\theta + 82.440)$$

TABLE 3

Porosity

Bottom Composition	Approximate Porosity	
Fine clay	0.93	(High Loss)
Medium clay	0.87	
Coarse clay	0.81	
Very fine silt	0.75	
Fine silt	0.69	
Medium silt	0.63	
Coarse silt	0.57	
Very fine sand	0.51	
Fine sand	0.45	
Medium sand	0.39	
Coarse sand	0.33	
Very coarse sand	0.27	(Low Loss)

TABLE 4
NEL and NUWC Equations

NEL Equation:

$$N_B = \left[6 + 22(P - .27) + 10 \log_{10} f\right] \cdot \left[\mathrm{TANH}\left((P/.24) \cdot (\theta/57.296)\right) + \left((1 - P/.24)/12.5\right) \cdot \left(\theta/90\right)^2\right]$$

NUWC Equation:

$$N_B = \left[3.7 + 17.5(P - .27)\right] \cdot \left[\mathrm{ABTMLOS}(f)\right] \cdot \left[\mathrm{TANH}\left((P/.24) \cdot (\theta/57.296)\right)^{1.5/P} + \left((1 - P/.27)/12.5\right) \cdot \left(\theta/90\right)^2\right]$$

where: N_B = Bottom Loss(dB)
f = Frequency(kHz)
P = Porosity
θ = Grazing Angle(deg)

NOTE: ABTMLOS is a function of frequency defined in Figure 7.

TABLE 5

Bottom Loss Model Options

Units Word	Sample Card	Model
	Fig. 1(a)	Inputted Table of Values.
AMOS	Fig. 1(b)	Least-Squares Fit to Modified AMOS data.
MGS1	Fig. 1(c)	Average Bottom Losses in MGS Acoustic Province 1.
MGS2	Fig. 1(d)	Average Bottom Losses in MGS Acoustic Province 2.
MGS3	Fig. 1(e)	Average Bottom Losses in MGS Acoustic Province 3.
MGS4	Fig. 1(f)	Average Bottom Losses in MGS Acoustic Province 4.
MGS5	Fig. 1(g)	Average Bottom Losses in MGS Acoustic Province 5.
NEL	Fig. 1(h)	Empirical Expression (NEL).
NUWC	Fig. 1(i)	Empirical Expression (NUWC).



Fig. 1 - BOTTOM LOSS Cards



Fig. 2 - FREQUENCY Card



Fig. 3 - AMOS Bottom Loss Curves



Fig. 4 - MGS Bottom Loss Curves



Fig. 5 - NEL Bottom Loss Curves
(Porosity = 0.60)



Fig. 6 - NEL Bottom Loss Curves
(Frequency = 1 kHz)

TABULATED FUNCTION
ABTMLOS

FREQUENCY	VALUE
.1	.157
.3	.45
.5	.67
1	1
1.5	1.18
2.5	1.43
4	1.70
6.5	2



Fig. 7 - Function ABTMLOS



Fig. 8 - NUWC Bottom Loss Curves
(Porosity = 0.60)



Fig. 9 - NUWC Bottom Loss Curves
(Frequency = 1 kHz)



Fig. 10 - AMOS, NEL, NUWC Bottom Loss Curves
(Frequency = 3.5 kHz, Porosity = 0.40)

APPENDIX
Subroutine Listings

```
C     SUBROUTINE  INPUT  FOR CONGRATS.                        CHIC WEINBERG
C     ---------------------------------------------------------------------
C
      SUBROUTINE  INPUT
C
C
      DIMENSION  TARGET(1010,3),ANGLES(1010),SONAR(5,2),SURFAC(5),
     1 SP(210,2),BOTTOM(5),BP(210,2),TOLERA(5),VP(210,2),TP(210,3)
      DIMENSION GROUP(11)/66HCOMMENTARGETSONAR SURFACBOTTOMVELOCITHERMA
     1PROCESEND   FREQUEMAXIMU/
      DIMENSION COORD(8)/48HRANGE DEPTH ANGLE LOSS  TOLERAAXES  PROFILP
     1HASE /
      DIMENSION UNITS(2,32)/
     1  1.09361111E-5, 6HCM     ,      1.09361111E-5, 6HCM/S  ,
     2  2.77777777E-5, 6HIN     ,      2.77777777E-5, 6HIN/S  ,
     3  0.33333333E-3, 6HFT     ,      0.33333333E-3, 6HFT/S  ,
     4  1.00000000E-3, 6HYD     ,      1.00000000E-3, 6HYD/S  ,
     5  1.09361111E-3, 6HM      ,      1.09361111E-3, 6HM/S   ,
     6  2.00000000E-3, 6HF      ,      2.00000000E-3, 6HF/S   ,
     7  0.33333333E-0, 6HKFT    ,      0.33333333E-0, 6HKFT/S ,
     8  1.00000000E-0, 6HKYD    ,      1.00000000E-0, 6HKYD/S ,
     9  1.09361111E-0, 6HKM     ,      1.09361111E-0, 6HKM/S  ,
     T  1.76000000E-0, 6HMI     ,      1.76000000E-0, 6HMI/S  ,
     1  2.02680000E-0, 6HN MI   ,      2.02680000E-0, 6HKNOTS ,
     2  1.74532925E-2, 6HDEG    ,      1.74532925E-2, 6HDEG/S ,
     3  1.00000000E-0, 6HRAD    ,      1.00000000E-0, 6HRAD/S ,
     4  6.28318531E-0, 6HCPS    ,      6.28318531E+3, 6HKCPS  ,
     5  1.00000000E-3, 6HMS     ,      1.00000000E-0, 6HSEC   ,
     6  1.00000000E-0, 6HC      ,      5.55555555E-1, 6HFAHR  /
      DIMENSION  TEST(12),FMT(12),DATA(2048),REVMAX(4)
      DIMENSION  CMMNT(12)/72H THERE SHOULD BE AT LEAST ONE COMMENT SET
     1.                                                                /
      DATA  PRINT/6HPRINT /,LINMAX/60/,ICMMNT/0/
      COMMON  TARGET,ANGLES,SONAR,SURFAC,SP,BOTTOM,BP,TOLERA,VP,
     1  REVMAX,PROCES
C
C
C
C     SET INITIAL CONDITIONS.
C     ----------------------
      ANGLES(1010) = 0.0
      TARGET(1010,1) = 0.0
      TARGET(1010,2) = 0.0
      IHEAD = 0
      LINES = LINMAX
      NPRINT = FLD(33,3,PROCES)
      NTAPE1 = FLD(30,3,PROCES)
      NTAPE2 = FLD(27,3,PROCES)
C
C
C
C     CHECK THE GROUP CODE.
C     --------------------
  100 READ 101, (TEST(J),J=1,9)
  101 FORMAT( 3(A6,A4), 3F10.5 )
      DO 110  I=1,11
      IF( TEST(1).GT.GROUP(I) .OR. TEST(1).LT.GROUP(I) )  GO TO 110
      IGROUP = I
```

```
      GO TO (1000,200,200,200,200,200,200,8000,9000,300,9200), IGROUP
  110 CONTINUE
C
C     THE GROUP CODE IS INCORRECT.
  120 FMT(1) = (+6HGROUP )
  130 PRINT 132, FMT(1)
  132 FORMAT( 10X 14HTHE FOLLOWING , A6, 18HCODE IS INCORRECT. )
  140 PRINT 142, ( TEST(J),J=1,9 )
  142 FORMAT( 24X 6A6, 3F12.5 )
  150 PRINT 152
  152 FORMAT( 10X 28HTHE PROGRAM CANNOT CONTINUE. )
      STOP 6
C
C
C     CHECK THE COORD CODE.
C     --------------------
  200 DO 210  I=1,8
      IF( TEST(3).GT.COORD(I) .OR. TEST(3).LT.COORD(I) )  GO TO 210
      ICOORD = I
      GO TO (300,300,300,500,300,300,600,500), ICOORD
  210 CONTINUE
C
C     THE COORD CODE IS INCORRECT.
      FMT(1) = (+6HCOORD )
      GO TO 130
C
C
C     CHECK THE UNITS CODE.
C     --------------------
  300 DO 310  I=1,30
      IF( TEST(5).GT.UNITS(2,I) .OR. TEST(5).LT.UNITS(2,I) )  GO TO 310
      IUNITS = I
      IF( IGROUP.EQ.10 )  GO TO 320
      IF( ICOORD.GE.6 )  GO TO 400
      IF( TEST(9) )  305,315,305
  305 NCARDS = (TEST(8)-TEST(7))/TEST(9) + 1.5
      IF( NCARDS*(1001-NCARDS) )  2100,2100,320
  310 CONTINUE
C
C     THE UNITS CODE IS INCORRECT.
      FMT(1) = (+6HUNITS )
      GO TO 130
C
C     PRINT THE DATA.
  315 TEST(8) = TEST(7)
      NCARDS = 1
  320 IF( LINES.LT.LINMAX-5 )  GO TO 330
      PRINT 322
  322 FORMAT( 1H1 )
      LINES = 1
      GO TO 340
  330 IF( IHEAD.EQ.1 )  GO TO 350
      PRINT 332
  332 FORMAT( // )
      LINES = LINES + 3
  340 PRINT 342
  342 FORMAT( 25X 73HGROUP     COORDINATE   UNITS            INITIAL
     1   FINAL       INCREMENT, / )
```

```
      LINES = LINES + 2
      IHEAD = 1
C
C     SET THE VARIABLE FORMAT.
  350 FMT(1) = (+6H( 24X )
      FMT(2) = (+6H6A6, 3)
      IF( UNITS(1,IUNITS).GT.2.0E-3 )  GO TO 360
      FMT(3) = (+6HF12.2))
      GO TO 370
  360 FMT(3) = (+6HF12.5))
  370 PRINT FMT, (TEST(J),J=1,9)
      LINES = LINES + 1
      TEST(7) = TEST(7) * UNITS(1,IUNITS)
      TEST(8) = TEST(8) * UNITS(1,IUNITS)
      TEST(9) = TEST(9) * UNITS(1,IUNITS)
      GO TO (380,2000,3000,4000,5000,6000,380,380,380,9100,100), IGROUP
C
C     THE CODES ARE INCONSISTENT.
  380 PRINT 382
  382 FORMAT( 10X 33HTHE ABOVE CODES ARE INCONSISTENT. )
      GO TO 150
C
C
C     CHECK THE AXES CARD.
C     -------------------
  400 IF( NTAPE1.EQ.0 )  NTAPE1=1
      IF( IUNITS.GT.4 )  GO TO 420
      IPAPER = 600 - 500 * ((IUNITS-1)/2)
      DO 410 J=1,3
      IF( TEST(J+6) * (100.0*UNITS(1,3)-TEST(J+6)*UNITS(1,IUNITS)) )
     1 420,410,410
  410 CONTINUE
      GO TO (420,420,420,4100,5100,6100,7100,420,420), IGROUP
C
C     THE AXES CARD IS INCORRECT.
  420 PRINT 422
  422 FORMAT( 10X 37HTHE FOLLOWING AXES CARD IS INCORRECT. )
      GO TO 140
C
C
C     READ THE LOSS TABLE.
C     -------------------
  500 IHEAD = 0
      LINES = LINES + 28
      IF( LINES.LT.LINMAX-5 )  GO TO 510
      PRINT 322
      LINES = 25
      GO TO 520
  510 PRINT 332
  520 IF( IGROUP.EQ.4 )  GO TO 540
      IF( IGROUP.NE.5 )  GO TO 120
      IF( ICOORD.EQ.4 )  GO TO 530
      TEST(12) = BPHASE(FREQ)
      GO TO 100
  530 FREQ = TARGET(1008,3) / UNITS(1,28)
      LINES = LINES + 5
      IF ( TEST(9) .LE. 0.0 )  TEST (9) = 500.0
      TEST(12) = BLOSS( FREQ,TEST(5),TEST(7),TEST(8),TEST(9) )
```

```
      GO TO 100
  540 IF( ICOORD.EQ.4 )  GO TO 550
      TEST(12) = SPHASE(FREQ)
      GO TO 100
  550 TEST(12) = SLOSS(FREQ)
      GO TO 100
C
C
C     CHECK THE NUMBER OF CARDS IN THE PROFILE.
C     ----------------------------------------
  600 NCARDS = TEST(7)
      IF( (NCARDS.GE.2).AND.(200.GE.NCARDS) )  GO TO 610
      PRINT 602, (TEST(J),J=1,4), NCARDS
  602 FORMAT( 10X 27HTHE NUMBER OF CARDS IN THE , 2(A6,A4),
     1 I10, 44H, EXCEEDS 200 CARDS OR IS LESS THAN 2 CARDS. )
      GO TO 150
C
C     PRINT THE HEADING.
  610 IHEAD = 0
      LINES = NCARDS/2 + 10 + LINES
      IF( LINES.LT.LINMAX-1 )  GO TO 620
      PRINT 322
      LINES = NCARDS/2 + 6
      GO TO 630
  620 PRINT 332
  630 PRINT 632, (TEST(J),J=1,4)
  632 FORMAT( 51X 2(A6,A4), /, 51X 17H-----------------, / )
C
C
C     CHECK THE UNITS CODE.
C     ---------------------
  700 READ 101, TEST(1),TEST(10),TEST(2),TEST(11),TEST(3),TEST(12)
      J = 1
  710 DO 730 I=1,32
      IF( TEST(J).GT.UNITS(2,I) .OR. TEST(J).LT.UNITS(2,I) )  GO TO 730
      IF( J.GT.1 )  GO TO 720
      IUNITS = I
      J = 2
      GO TO 710
  720 IF (J .GT. 2)  GO TO 725
      JUNITS = I
      J = 3
      IF (IGROUP .EQ. 7)  GO TO 710
      GO TO 740
  725 KUNITS = I
      GO TO 740
  730 CONTINUE
C
C     THE UNITS CODE IS INCORRECT.
      FMT(1) = (+6HUNITS )
      PRINT 132, FMT(1)
      PRINT 142, TEST(1),TEST(3),TEST(2),TEST(4)
      GO TO 150
C
C     SET THE VARIABLE FORMAT.
  740 FMT(1) = (+6H( 2(   )
      FMT(2) = (+6H I15, )
      IF( UNITS(1,IUNITS).GT.2.0E-3 )  GO TO 750
```

```
      FMT(3) = (+6HF14.2,)
      GO TO 760
  750 FMT(3) = (+6HF14.5,)
  760 IF (IGROUP .EQ. 7)  GO TO 763
      LUNITS = JUNITS
      GO TO 765
  763 LUNITS = KUNITS
  765 IF ( UNITS(1,LUNITS) .GT. 2.0E-3 )  GO TO 770
      FMT(4) = (+6HF16.2,)
      GO TO 780
  770 FMT(4) = (+6HF16.5,)
  780 FMT(5) = (+6H15X) ))
      GO TO (380,380,380,4200,5200,6200,7200,380,380), IGROUP
C
C
C
C     READ AND WRITE A COMMENT CARD.
C     ------------------------------
 1000 READ 1001, (TEST(J),J=1,12)
 1001 FORMAT( 12A6 )
 1002 FORMAT( 10X 12A6 )
      IF( LINES.LT.LINMAX )  GO TO 1005
      PRINT 322
      LINES = 1
 1005 PRINT 1002, (TEST(J),J=1,12)
      LINES = LINES + 1
      ICMMNT = ICMMNT + 1
      IF( ICMMNT.GT.1 )  GO TO 100
      DO 1010  J=1,12
 1010 CMMNT(J) = TEST(J)
      GO TO 100
C
C
C
C     CONVERT AND STORE THE TARGET DATA.
C     ----------------------------------
 2000 IF( ICOORD.GT.2 )  GO TO 380
      IF( NTAPE2.EQ.0 )  NTAPE2=1
      TARGET(1004,ICOORD) = UNITS(1,IUNITS)
      TARGET(1005,ICOORD) = UNITS(2,IUNITS)
      N = TARGET(1010,ICOORD)
      NCARDS = N + NCARDS
      IF( NCARDS.GT.1000 )  GO TO 2100
      TARGET(1010,ICOORD) = NCARDS
      N = N + 1
      DO 2010  I=N,NCARDS
 2010 TARGET(I,ICOORD) = TEST(7) + (I-N)*TEST(9)
      GO TO 100
C
C     THE NUMBER OF ENTRIES HAS BEEN EXCEEDED.
 2100 PRINT 2102, GROUP(IGROUP), COORD(ICOORD)
 2102 FORMAT( 10X 14HTHE NUMBER OF , A6, 1XA5,
     1  23HS EXCEEDS 1000 ENTRIES. )
      GO TO 150
C
C
C
C     CONVERT AND STORE THE SONAR DATA.
```

```
C      -------------------------------
 3000  IF( ICOORD-3 )  3100,3200,380
 3100  SONAR(1,ICOORD) = TEST(7)
       SONAR(2,ICOORD) = TEST(8)
       SONAR(3,ICOORD) = TEST(9)
       SONAR(4,ICOORD) = UNITS(1,IUNITS)
       SONAR(5,ICOORD) = UNITS(2,IUNITS)
       GO TO 100
C
C      STORE THE SONAR ANGLE DATA.
 3200  IF( ABS(TEST(7)).GT.1.5707 .OR. TEST(8).GT.1.5707 )  GO TO 3300
       ANGLES(1004) = UNITS(1,IUNITS)
       ANGLES(1005) = UNITS(2,IUNITS)
       N = ANGLES(1010)
       NCARDS = N + NCARDS
       IF( NCARDS.GT.1000 )  GO TO 2100
       ANGLES(1010) = NCARDS
       N = N + 1
       DO 3210  I=N,NCARDS
 3210  ANGLES(I) = TEST(7) + (I-N)*TEST(9)
       GO TO 100
C
C      THE SONAR ANGLES EXCEED THEIR BOUNDS.
 3300  PRINT 3302
 3302  FORMAT(10X50HALL SONAR ANGLES MUST LIE BETWEEN -90 AND +90 DEG.)
       GO TO 150
C
C
C
C      CONVERT AND STORE THE SURFACE DATA.
C      ----------------------------------
 4000  IF( ICOORD.NE.2 )  GO TO 380
       SURFAC(1) = TEST(7)
       SURFAC(2) = TEST(8)
       SURFAC(3) = TEST(9)
       SURFAC(4) = UNITS(1,IUNITS)
       SURFAC(5) = UNITS(2,IUNITS)
       GO TO 100
C
C      STORE THE SURFACE AXES DATA.
 4100  SP(207,1) = 2.777777777E-4/UNITS(1,IUNITS)
       SP(207,2) = SP(207,1)
       SP(208,1) = TEST(7)
       SP(208,2) = TEST(8)
       GO TO 5100
C
C
C      READ AND PRINT THE SURFACE PROFILE.
C      ----------------------------------
 4200  SP(204,1) = UNITS(1,IUNITS)
       SP(205,1) = UNITS(2,IUNITS)
       SP(204,2) = UNITS(1,JUNITS)
       SP(205,2) = UNITS(2,JUNITS)
       SP(210,1) = NCARDS
       SP(210,2) = TEST(5)
       SURFAC(4) = 0.0
       PRINT 4202, SP(205,1),SP(205,2),SP(205,1),SP(205,2)
 4202  FORMAT( 2(12X 16HCARD       RANGE-, A6, 9H   DEPTH-, A6, 11X), / )
```

```
      READ 4212, (SP(N,1),SP(N,2),N=1,NCARDS)
 4212 FORMAT( 2F10.5 )
      J = ( NCARDS + 1 ) / 2
      DO 4220  L=1,J
      N = L + J
      IF( N.LE.NCARDS )  GO TO 4220
      PRINT FMT, L,SP(L,1),SP(L,2)
      GO TO 4230
 4220 PRINT FMT, L,SP(L,1),SP(L,2), N,SP(N,1),SP(N,2)
C
C     CONVERT AND CHECK THE PROFILE.
 4230 DO 4260  N=1,NCARDS
      SP(N,1) = SP(N,1) * SP(204,1)
      SP(N,2) = SP(N,2) * SP(204,2)
C
C     IS THE RANGE INCREASING.
      IF( N.EQ.1 )  GO TO 4260
      IF( SP(N,1).GT.SP(N-1,1) )  GO TO 4260
      NCARDS = N
      GO TO 4270
 4260 CONTINUE
      GO TO 100
C
C     A CARD IS INCORRECT OR OUT OF ORDER.
 4270 PRINT 4272, NCARDS
 4272 FORMAT( 10X 11HCARD NUMBER, I4, 13HIS INCORRECT. )
      GO TO 150
C
C
C
C     CONVERT AND STORE THE BOTTOM DATA.
C     ----------------------------------
 5000 IF( ICOORD.NE.2 )  GO TO 380
      BOTTOM(1) = TEST(7)
      BOTTOM(2) = TEST(8)
      BOTTOM(3) = TEST(9)
      BOTTOM(4) = UNITS(1,IUNITS)
      BOTTOM(5) = UNITS(2,IUNITS)
      GO TO 100
C
C     STORE THE BOTTOM AXES DATA.
 5100 BP(207,1) = 2.777777777E-4/UNITS(1,IUNITS)
      BP(207,2) = BP(207,1)
      BP(208,1) = TEST(7)
      BP(208,2) = TEST(8)
      GO TO 100
C
C
C     READ AND PRINT THE BOTTOM PROFILE.
C     ----------------------------------
 5200 BP(204,1) = UNITS(1,IUNITS)
      BP(205,1) = UNITS(2,IUNITS)
      BP(204,2) = UNITS(1,JUNITS)
      BP(205,2) = UNITS(2,JUNITS)
      BP(210,1) = NCARDS
      BP(210,2) = TEST(5)
      BOTTOM(4) = 0.0
      PRINT 4202, BP(205,1),BP(205,2),BP(205,1),BP(205,2)
```

```
      READ 4212, (BP(N,1),BP(N,2),N=1,NCARDS)
      J = ( NCARDS + 1 ) / 2
      DO 5220  L=1,J
      N = L + J
      IF( N.LE.NCARDS )  GO TO 5220
      PRINT FMT, L,BP(L,1),BP(L,2)
      GO TO 5230
 5220 PRINT FMT, L,BP(L,1),BP(L,2), N,BP(N,1),BP(N,2)
C
C     CONVERT AND CHECK THE PROFILE.
 5230 DO 5260  N=1,NCARDS
      BP(N,1) = BP(N,1) * BP(204,1)
      BP(N,2) = BP(N,2) * BP(204,2)
C
C     IS THE RANGE INCREASING.
      IF( N.EQ.1 )  GO TO 5260
      IF( BP(N,1).GT.BP(N-1,1) )  GO TO 5260
      NCARDS = N
      GO TO 4270
 5260 CONTINUE
      GO TO 100
C
C
C
C     CONVERT AND STORE THE VELOCITY DATA.
C     ------------------------------------
 6000 IF( ICOORD.NE.5 )  GO TO 380
      TOLERA(1) = TEST(7)
      TOLERA(2) = TEST(8)
      TOLERA(3) = TEST(9)
      TOLERA(4) = UNITS(1,IUNITS)
      TOLERA(5) = UNITS(2,IUNITS)
      GO TO 100
C
C     STORE THE VELOCITY AXES DATA.
 6100 VP(207,1) = 2.777777777E-4/UNITS(1,IUNITS)
      VP(207,2) = VP(207,1)
      VP(208,1) = TEST(7)
      VP(208,2) = TEST(8)
      GO TO 100
C
C
C     READ AND PRINT THE VELOCITY PROFILE.
C     ------------------------------------
 6200 VP(204,1) = UNITS(1,IUNITS)
      VP(205,1) = UNITS(2,IUNITS)
      VP(204,2) = UNITS(1,JUNITS)
      VP(205,2) = UNITS(2,JUNITS)
      VP(210,1) = NCARDS
      VP(210,2) = TEST(5)
      PRINT 6202, VP(205,1),VP(205,2),VP(205,1),VP(205,2)
 6202 FORMAT(2(12X 16HCARD      DEPTH-, A6, 10H VELOCITY-, A6, 10X), /)
      READ 4212, (VP(N,1),VP(N,2),N=1,NCARDS)
 6210 J = ( NCARDS + 1 )/2
      DO 6220  L=1,J
      N = L + J
      IF( N.LE.NCARDS )  GO TO 6220
      PRINT FMT, L,VP(L,1),VP(L,2)
```

```
      GO TO 6230
 6220 PRINT FMT, L,VP(L,1),VP(L,2), N,VP(N,1),VP(N,2)
C
C     CONVERT AND CHECK THE PROFILE.
 6230 DO 6260  N=1,NCARDS
      VP(N,1) = VP(N,1) * VP(204,1)
      VP(N,2) = VP(N,2) * VP(204,2)
      IF( VP(N,2).GT.1.55 .AND. 1.77.GT.VP(N,2) )  GO TO 6240
C
C     THE VELOCITY EXCEEDS ITS BOUNDS.
      PRINT 6232, N
 6232 FORMAT( 10X 28HVELOCITY PROFILE CARD NUMBER, I4,
    1  20H EXCEEDS ITS BOUNDS. )
      GO TO 150
C
C     IS THE DEPTH INCREASING.
 6240 IF( N.EQ.1 )  GO TO 6260
      IF( VP(N,1).GT.VP(N-1,1) )  GO TO 6260
      NCARDS = N
      GO TO 4270
 6260 CONTINUE
      GO TO 100
C
C
C
C     CONVERT AND STORE TEMPERATURE DATA.
C     ----------------------------------
C
C
C     STORE THE THERMAL AXES DATA.
 7100 TP(207,1) = 2.777777777E-4/UNITS(1,IUNITS)
      TP(207,2) = TP(207,1)
      TP(207,3) = TP(207,1)
      TP(208,1) = TEST (7)
      TP(208,2) = TEST (8)
      TP(208,3) = TEST (9)
      GO TO 100
C
C     READ AND PRINT THE THERMAL PROFILE.
C     ----------------------------------
C
 7200 VP(204,1) = UNITS(1,IUNITS)
      VP(205,1) = UNITS(2,IUNITS)
      VP(204,2) = UNITS(1,KUNITS)
      VP(205,2) = UNITS(2,KUNITS)
      VP(210,1) = NCARDS
      VP(210,2) = TEST(5)
      TP(204,1) = UNITS(1,IUNITS)
      TP(205,1) = UNITS(2,IUNITS)
      TP(204,2) = UNITS(1,JUNITS)
      TP(205,2) = UNITS(2,JUNITS)
      TP(204,3) = UNITS(1,9)
      TP(205,3) = UNITS(1,10)
      TP(210,1) = NCARDS
      TP(210,2) = TEST(8)
      TP(210,3) = TEST (9)
      PRINT 7202, TP(205,1),TP(205,2),TP(205,1),TP(205,2)
 7202 FORMAT(2(6X,16HCARD      DEPTH-,A6,12HTEMPERATURE-,A6,14HSALINITY
```

```
     1-/1000,6X),/)
      READ 7205, (TP(N,1),TP(N,2),TP(N,3),N=1,NCARDS)
 7205 FORMAT(3F10.5)
C
C     TEST WHICH SALINITY INPUT TO USE.
C
      SAL = 0.0
      DO 7207 I = 1,NCARDS
      IF (  TP(I,3) )  7208,     ,     ,
      SAL = SAL + TP(I,3)
      IF (SAL .GT. 0.0)  GO TO 7215
 7207 CONTINUE
      GO TO 7210
 7208 PRINT 7237, I
      PRINT 7209
 7209 FORMAT (10X,36HNEGATIVE SALINITY IS NOT VALID DATA.)
      GO TO 150
 7210 DO 7212 I = 1,NCARDS
      TP(I,3) = TP(210,3)
 7212 CONTINUE
C
C     PRINT THE TEMPERATURE PROFILE.
C
 7215 J = ( NCARDS + 1 )/2
      DO 7220 L=1,J
      N = L + J
      IF (N .LE. NCARDS)  GO TO 7220
      PRINT 7225, L,TP(L,1),TP(L,2),TP(L,3)
      GO TO 7226
 7220 PRINT 7225, L,TP(L,1),TP(L,2),TP(L,3),N,TP(N,1),TP(N,2),TP(N,3)
 7225 FORMAT(2(6X,I3,6X,F8.2,8X,F7.2,11X,F7.2,10X))
 7226 PRINT 7227, TP(210,2)
 7227 FORMAT(//,45X,12HLATITUDE =  ,F10.5,8H DEGREES,//)
C
C     CONVERT AND CHECK TEMPERATURE PROFILE
C
 7230 DO 7260 N=1,NCARDS
      IF (JUNITS .EQ. 31)  GO TO 7235
      TP(N,2) = (TP(N,2) - 32.) * TP(204,2)
 7235 IF (TP(N,2) .GT.-3.0 .AND. 35.0 .GT. TP(N,2))  GO TO 7250
C
C     THE TEMPERATURE EXCEEDS NEPTUNIAN BOUNDS.
C
      PRINT 7237, N
 7237 FORMAT(10X,31HTEMPERATURE PROFILE CARD NUMBER,I4,19HEXCEEDS ITS B
     1OUNDS.)
C
C     CHECK IF SALINITY IS WITHIN NEPTUNIAN BOUNDS.
C
 7250 IF (SAL .GT. 0.0)  GO TO 7254
      K = N
      IF (K .GT. 1 )  GO TO 7260
      IF (TP(210,3) .GE. 0.0 .AND. 43.0 .GT. TP(210,3))  GO TO 7260
      GO TO 7256
 7254 IF (TP(N,3) .GE. 0.0 .AND. 43.0 .GT. TP(210,3))  GO TO 7260
C
C     THE SALINITY EXCEEDS NEPTUNIAN BOUNDS.
C
```

```
 7255  PRINT 7257
       PRINT 7237, N
       GO TO 7260
 7256  PRINT 7257
 7257  FORMAT(10X,40HSALINITY IS NOT WITHIN NEPTUNIAN BOUNDS.)
 7260  CONTINUE
       CALL BT(TP)
C
C      PRINT THE HEADING
C
       IHEAD = 0
       LINES = NCARDS/2 + 10 + LINES
       IF (LINES .LT. LINMAX - 1)  GO TO 7320
       PRINT 322
       LINES = NCARDS/2 + 6
       GO TO 7330
 7320  PRINT 332
 7330  PRINT 7335
 7335  FORMAT ( 51X,17HVELOCITY PROFILE ,/,51X,17H----------------,/)
C
C      PRINT THE VELOCITY PROFILE.
C
       PRINT 6202, VP(205,1),VP(205,2),VP(205,1),VP(205,2)
       GO TO 6210
C
C
C
C      PROCESS CONGRATS.
C      ----------------
 8000  IF( PRINT.LT.TEST(3) .OR. PRINT.GT.TEST(3) )  GO TO 8010
       NPRINT = 1
       GO TO 8020
 8010  NPRINT = 0
 8020  FLD(27,9,PROCES) = NPRINT + 8 * (NTAPE1 + 8*NTAPE2)
C
C      IS THERE AT LEAST ONE COMMENT CARD.
       IF( ICMMNT.GT.0 )  GO TO 8100
       PRINT 1001, (CMMNT(J),J=1,12)
       LINES = LINES + 1
C
C      INITIALIZE THE PLOTTER TAPE.
 8100  IF( NTAPE1.NE.1 )  GO TO 8200
       CALL  PLOTS( DATA,2048,1 )
       CALL  PLOT(0.0,0.0,IPAPER)
       CALL  PLOT(5.0,0.0,-3)
       CALL  SYMBOL(0.0,0.0,0.14,CMMNT(1),90.0,72)
       CALL  PLOT(5.0,0.0,-3)
       NTAPE1 = 2
C
C      SORT THE TARGET COORDINATES.
 8200  DO 8260  J=1,2
       N = TARGET(1010,J)
       IF( N.LT.2 )  GO TO 8260
       L = 1
 8210  I = L + 1
 8220  IF( TARGET(L,J)-TARGET(I,J) )  8250,8230,8240
 8230  TARGET(I,J) = TARGET(N,J)
       N = N - 1
```

```
      GO TO 8250
 8240 TARGET(N+1,J) = TARGET(I,J)
      TARGET(I,J) = TARGET(L,J)
      TARGET(L,J) = TARGET(N+1,J)
 8250 I = I + 1
      IF( I.LE.N )  GO TO 8220
      L = L + 1
      IF( L.LT.N )  GO TO 8210
      TARGET(1010,J) = N
 8260 CONTINUE
      CALL BTPLOT(TP)
C
C     SET THE MAXIMUM REVERSAL INCREMENT.
      IF( REVMAX(2).GT.REVMAX(1) )  GO TO 8270
      REVMAX(3) = 0.0
      GO TO 8300
 8270 N = TARGET(1010,1)
      REVMAX(3) = (TARGET(N,1)-TARGET(1,1))/(REVMAX(2)-REVMAX(1))
C
C     SORT THE SONAR ANGLES.
 8300 N = ANGLES(1010)
      IF( N.LT.1 )  GO TO 8400
      DO 8305  L=1,N
      IF( ABS(ANGLES(L)).GT.1.0E-4 )  GO TO 8305
      ANGLES(L) = 0.0
 8305 CONTINUE
      IF( N.LT.2 )  GO TO 8400
      L = 1
 8310 I = L + 1
 8320 IF( ANGLES(L)-ANGLES(I) )  8350,8330,8340
 8330 ANGLES(I) = ANGLES(N)
      N = N - 1
      GO TO 8350
 8340 ANGLES(N+1) = ANGLES(I)
      ANGLES(I) = ANGLES(L)
      ANGLES(L) = ANGLES(N+1)
 8350 I = I + 1
      IF( I.LE.N )  GO TO 8320
      L = L + 1
      IF( L.LT.N )  GO TO 8310
      ANGLES(1010) = N
 8400 RETURN
C
C
C
C     TERMINATE CONGRATS.
C     -------------------
 9000 PRINT 9002
 9002 FORMAT( 1H1, 9X 28HCONGRATS HAS BEEN COMPLETED. , /, 1H1 )
 9030 IF( NTAPE2.EQ.0 )  GO TO 9090
      END FILE 2
 9090 STOP 5
C
C
C
C     READ MISCELLANEOUS DATA.
C     ------------------------
C
```

```
C       READ THE FREQUENCY.
 9100   TARGET(1008,3) = TEST(7)
        IF( (IUNITS-27)*(IUNITS-28) )  380,9110,380
 9110   TARGET(1008,1) = ATTEN(TARGET(1008,3))
        PRINT 9112, TARGET(1008,1)
 9112   FORMAT( 24X, 12HATTENUATION , 12X 6HDB/KYD,  6X F12.5 )
        LINES = LINES + 1
        GO TO 100
C
C       READ THE MAXIMUM NUMBER OF REVERSALS.
 9200   REVMAX(1) = TEST(7)
        REVMAX(2) = MAX(TEST(7),TEST(8))
        REVMAX(4) = TEST(9)
        TEST(8) = REVMAX(2)
        IUNITS = 15
        IF( REVMAX(2).LT.201.0 )  GO TO 320
        PRINT 9202
 9202   FORMAT( 10X45HTHE MAXIMUM NUMBER OF REVERSALS EXCEEDS 200.0 )
        GO TO 140
        END
```

```
C     FUNCTION  BLOSS  AND  SLOSS  FOR CONGRATS.            CHIC WEINBERG
C     ---------------------------------------------------------------------
C
C
C
C     READ AND PRINT THE LOSS TABLE.
C     ----------------------------
C
      FUNCTION BLOSS( FREQ,TEST,POR,DBMIN,DBMAX )
      DOUBLE PRECISION  R,Z,RB,ZB,COS2B,RAD2B,SIN2B
      DIMENSION  DB(92,4)/276*0.0,92*3.14159265/
      DIMENSION  TYPE(10)/60H      AMOS  MGS1  MGS2  MGS3  MGS4  MGS5
     1NEL   NUWC  OTHERS
      PRINT 2
    2 FORMAT( 54X 11HBOTTOM LOSS, /, 54X 11H-----------, / )
      J = 1
      DO 10 I=1,10
      IF( TEST.GT.TYPE(I) .OR. TEST.LT.TYPE(I) ) GO TO 10
      ITYPE = I - 1
      IF( ITYPE .LE. 0 ) GO TO 100
      CALL BTMLOS( FREQ,DB,ITYPE,POR,DBMIN,DBMAX,$110 )
   10 CONTINUE
C
      ENTRY  SLOSS( FREQ )
      PRINT 12
   12 FORMAT( 54X 12HSURFACE LOSS, /, 54X 12H------------, / )
      J = 2
      GO TO 100
C
      ENTRY  BPHASE(FREQ)
      PRINT 22
   22 FORMAT( 54X 12HBOTTOM PHASE, /, 54X 12H------------, / )
      J = 3
      GO TO 100
C
      ENTRY  SPHASE(FREQ)
      PRINT 32
   32 FORMAT( 54X 13HSURFAC  PHASE, /, 54X 13H-------------, / )
      J = 4
  100 READ 101, (DB(I,J),I=1,91)
  101 FORMAT( 10F5.1 )
  110 DB(92,J) = DB(91,J)
      IF( J.GT.2 )  GO TO 150
      PRINT 112
  112 FORMAT( 5(24H     ANGLE     LOSS      ),
     1      /, 5(24H     DEG       DB        ), / )
      GO TO 160
  150 PRINT 152
  152 FORMAT( 5(24H     ANGLE    PHASE      ),
     1      /, 5(24H     DEG      DEG       ), / )
  160 DO 170  I=1,18
      I1 = I - 1
      I2 = I1 + 19
      I3 = I2 + 18
      I4 = I3 + 18
      I5 = I4 + 18
  170 PRINT 172, I1,DB(I1+1,J), I2,DB(I2+1,J), I3,DB(I3+1,J),
     1  I4,DB(I4+1,J), I5,DB(I5+1,J)
```

```
  172 FORMAT( I8,F11.2, I13,F11.2, I13,F11.2, I13,F11.2, I13,F11.2 )
      I1 = 18
      PRINT 172, I1,DB(I1+1,J)
      IF ( ITYPE .GT. 0 )  PRINT 174, TYPE(ITYPE+1)
  174 FORMAT(//,40X,'METHOD OF BOTTOM LOSS COMPUTATION:  ',A6)
      IF ( POR .GT. 0.0 )  PRINT 175, POR
  175 FORMAT ( 52X,'POROSITY = ',F4.2)
      IF (DBMIN .GT. 0.0 .OR. DBMIN .LT. 0.0 ) PRINT 176, DBMIN
  176 FORMAT ( 41X,'MINIMUM LIMIT FOR BOTTOM LOSS = ',F6.2)
      IF ( DBMAX .GT. 0.0 .AND. DBMAX .LT. 500.0 ) PRINT 177, DBMAX
  177 FORMAT ( 41X,'MAXIMUM LIMIT FOR BOTTOM LOSS = ',F6.2)
      FREQUE = FREQ
      IF( J.LE.2 )  GO TO 190
C
C     CONVERT THE PHASE TO RADIANS.
      DO 180  I=1,92
  180 DB(I,J) = DB(I,J) * 1.74532925E-2
  190 RETURN
C
C
C
C     USE THE LOSS TABLE.
C     ------------------
C
      ENTRY  BOTLOS(R,Z,RB,ZB,PHASE)
      J = 1
      GO TO 200
C
      ENTRY  SURLOS(R,Z,RB,ZB,PHASE)
      J = 2
  200 IF( R*RB+Z*ZB )  220,210,220
  210 ANGLE = 90.0
      GO TO 230
  220 ANGLE = 57.29578D+0 * ATAN((R*ZB-Z*RB)/(R*RB+Z*ZB))
  230 I = ABS(ANGLE) + 1.0
      ANGLEI = I - 1
      BLOSS = DB(I,J) + (DB(I+1,J)-DB(I,J)) * (ABS(ANGLE)-ANGLEI)
      PHASE = DB(I,J+2) + (DB(I+1,J+2)-DB(I,J+2)) * (ABS(ANGLE)-ANGLEI)
C
C
C
C     COMPUTE THE DIRECTION NUMBERS OF THE REFLECTED RAY.
C     ---------------------------------------------------
      RAD2B = RB*RB + ZB*ZB
      COS2B = (RB+ZB)*(RB-ZB) / RAD2B
      SIN2B = 2.0D+0*RB*ZB/RAD2B
      RAD2B = COS2B*R + SIN2B*Z
      Z     = SIN2B*R - COS2B*Z
      R     = RAD2B
      RETURN
      END
```

```
      SUBROUTINE BTMLOS (FREQ,DB,ITYPE,POR,DBMIN,DBMAX,$)
C
      DIMENSION DB(91)
      DIMENSION FR(8)/0.1,0.3,0.5,1.0,1.5,2.5,4.0,6.5/
      DIMENSION FVAL(8)/0.157,0.45,0.67,1.0,1.18,1.43,1.7,2.0/
      GO TO (100,1000,2000,3000,4000,5000,6000,7000,9000),ITYPE
C
C     ------------------------------------------------
C     BOTTOM LOSS FROM AMOS DATA.
C     ------------------------------------------------
C
C
  100 FREQUE = FREQ
      FRQLG2 = LOG(FREQUE)/LOG(2.0)
C
C     GRAZING ANGLES ARE 0 THROUGH 14.
C
      F = FRQLG2
      FSQ = F*F
      FCUBE = FSQ*F
      F4PWR = FCUBE*F
      DB(1) = 0.0
      DO 110 I = 1,14
      THETA = I
      TH = 0.2*(THETA - 10.0)
  110 DB(I+1) = 2.96026 + 1.26990*F + 1.60430*TH + 0.25257*FSQ
     1          + 0.66161*F*TH + 0.06072*TH*TH + 0.02000*FCUBE
     2          + 0.09833*FSQ*TH - 0.01888*F*TH*TH - 0.02619*TH**3
     3          - 0.00303*F4PWR + 0.00778*FCUBE*TH - 0.00833*FSQ*TH*TH
     4          - 0.01250*F*TH**3 - 0.01310*TH**4
C
C     TEST FOR FREQUENCY GREATER THAN 1.0 KC.
C
      IF (FREQUE .GT. 1.0)  GO TO 200
C
C     FREQUENCY IS NOT GREATER THAN 1.0 KC
C     ------------------------------------
C
      F = 2.0*FRQLG2 + 3.0
      FSQ = F*F
      FCUBE = FSQ*F
      F4PWR = FCUBE*F
C
C     GRAZING ANGLES ARE 15 THROUGH 39
C
      DO 120 I=15,39
      THETA = I
      TH = 0.2*(THETA - 30.0)
  120 DB(I+1) = 4.35024 + 1.17091*F + 0.46874*TH + 0.08272*FSQ
     1          + 0.17115*F*TH - 0.07547*TH*TH - 0.00386*FCUBE
     2          + 0.01295*FSQ*TH - 0.01678*F*TH*TH - 0.00555*TH**3
     3          - 0.00003*F4PWR - 0.00193*FCUBE*TH - 0.00074*FSQ*TH*TH
     4          - 0.00224*F*TH**3
C
C     GRAZING ANGLES ARE 40 THROUGH 60
C
      DO 130 I=40,60
      THETA = I
```

```
      TH = 0.2*(THETA - 50.0)
  130 DB(I+1) = 4.84820 + 1.46858*F - 0.11457*TH + 0.11036*FSQ
     1           - 0.01238*F*TH - 0.00953*TH*TH - 0.01000*FCUBE
     2           - 0.00375*FSQ*TH - 0.00929*F*TH*TH + 0.00833*TH**3
     3           - 0.00042*FCUBE*TH - 0.00268*FSQ*TH*TH + .00830*F*TH**3
     4           -0.00208*TH**4
C
C     GRAZING ANGLES ARE 61 THROUGH 90
C
      F = FRQLG2
      DO 140 I=61,90
  140 DB(I+1) = DB(I) - 0.085667 - 0.025434*F
      GO TO 9000
C
C     FREQUENCY IS GREATER THAN 1.0 KC.
C     ----------------------------------
C
  200 F = FRQLG2 - 1
      FSQ = F*F
      FCUBE = FSQ*F
      F4PWR = FCUBE*F
C
C     GRAZING ANGLES ARE 15 THROUGH 39
C
      DO 220 I=15,39
      THETA = I
      TH = 0.2*(THETA - 30.0)
  220 DB(I+1) = 12.25083 + 3.88808*F + 1.14802*TH + 0.16312*FSQ
     1           + 0.04386*F*TH - 0.18835*TH*TH - 0.02857*FCUBE
     2           - 0.08214*FSQ*TH - 0.05952*F*TH*TH + 0.00667*TH**3
     3           + 0.01071*F4PWR - 0.00179*FCUBE*TH + 0.00476*FSQ*TH*TH
     4           + 0.01389*F*TH**3 - 0.00348*TH**4
C
C     GRAZING ANGLES ARE 40 THROUGH 60
C
      DO 230 I=40,60
      THETA = I
      TH = 0.2*(THETA - 50.0)
  230 DB(I+1) = 13.59067 + 3.74190*F - 0.25928*TH + 0.13266*FSQ
     1           - 0.03764*F*TH - 0.08816*TH*TH - 0.01333*FCUBE
     2           + 0.01214*FSQ*TH - 0.00959*F*TH*TH + 0.00500*TH**3
     3           - 0.02000*F4PWR - 0.00167*FCUBE*TH + 0.00867*FSQ*TH*TH
     4           - 0.00167*F*TH**3 + 0.00083*TH**4
C
C     GRAZING ANGLES ARE 61 THROUGH 90
C
      F = FRQLG2
      DO 240 I=61,90
  240 DB(I+1) = DB(I) - 0.126002 - 0.000333*F + 0.034002/F
      GO TO 9000
C
C
C     ----------------------------------------------------
C     BOTTOM LOSS FOR ACOUSTIC PROVINCES.  MGS 1 - 5.
C     ----------------------------------------------------
C
C
C
C     BOTTOM LOSS FOR ACOUSTIC PROVINCE MGS 1.
```

```
C       -----------------------------------------
C
 1000   DO 1020 I=0,90
        TH = I
 1020   DB(I+1) = 2.2435*LOG(0.1260*TH + 1.496)
        GO TO 9000
C
C       BOTTOM LOSS FOR ACOUSTIC PROVINCE MGS 2.
C       -----------------------------------------
C
 2000   DO 2020 I=0,90
        TH = I
 2020   DB(I+1) = 3.4315*LOG(0.1056*TH + 2.842)
        GO TO 9000
C
C       BOTTOM LOSS FOR ACOUSTIC PROVINCE MGS 3.
C       -----------------------------------------
C
 3000   DO 3020 I=0,90
        TH = I
 3020   DB(I+1) = 2.4910*LOG(0.8864*TH + 10.526)
        GO TO 9000
C
C       BOTTOM LOSS FOR ACOUSTIC PROVINCE MGS 4.
C       -----------------------------------------
C
 4000   DO 4020 I=0,90
        TH = I
 4020   DB(I+1) = 2.8377*LOG(1.8754*TH + 15.685)
        GO TO 9000
C
C       BOTTOM LOSS FOR ACOUSTIC PROVINCE MGS 5.
C       -----------------------------------------
C
 5000   DO 5020 I=0,90
        TH = I
 5020   DB(I+1) = 2.4036*LOG(20.5760*TH + 82.440)
        GO TO 9000
C
C
C       -----------------------------------------------
C       NAVAL ELECTRONIC LABORATORY EQUATIONS.
C       LOSS DEPENDENT ON POROSITY OF BOTTOM.
C       -----------------------------------------------
C
C
 6000   FREQUE = FREQ
        F = 10.0*LOG10(FREQUE)
        POROS = POR
        DO 6020 I=0,90
        TH = I
 6020   DB(I+1) = (6.0 + 22.0*(POROS - 0.27) + F) * (TANH((POROS/0.24) *
     1            (TH/57.295779)) + (1.0 - POROS/0.24)/12.5 *
     2            TH*TH/8100.0)
        GO TO 9000
C
C
C       --------------------------------------------
```

```
C       NAVAL UNDERSEA WARFARE CENTER EQUATIONS.
C       LOSS DEPENDENT ON POROSITY OF BOTTOM.
C       ------------------------------------------
C
 7000   FREQUE = FREQ
        IF ( FREQUE - 0.1 ) 7030, 7030,
        IF ( FREQUE - 6.5 )      ,      , 7040
        DO 7020 I=1,8
        IF ( (FREQUE - FR(I))*(FR(I+1) - FREQUE) ) 7020,      ,
        FUNU = FVAL(I) + (FREQUE - FR(I))*(FVAL(I+1) - FVAL(I))/(FR(I+1)
     1         - FR(I) )
        GO TO 7100
 7020   CONTINUE
 7030   FUNU = FVAL(1)
        GO TO 7100
 7040   FUNU = FVAL(8)
        GO TO 7100
 7100   POROS = POR
        FRQFUN = (3.7 + 17.5*(POROS - 0.27) )*FUNU
        PORTAN = POROS/13.75104
        SUPER = 1.5/POROS
        PORDIV = (1 - POROS/0.27)/101250.0
        DO 7120 I=0,90
        TH = I
 7120   DB(I+1) = FRQFUN*(TANH( (PORTAN*TH)**SUPER ) + PORDIV*TH*TH )
        GO TO 9000
C
C       ------------------------------------------------
C       CHECK LOSS AGAINST MINIMUM AND MAXIMUM
C       ------------------------------------------------
C
C
 9000   DO 9010 I=1,91
        IF ( DB(I) .LT. DBMIN )  DB(I) = DBMIN
        IF ( DB(I) .GT. DBMAX )  DB(I) = DBMAX
 9010   CONTINUE
 9900   RETURN 7
        END
```